3. Versengés, a Cournot-duopólium

Ebben a fejezetben a Cournot-duopólium legegyszerűbb modelljeit vizsgáljuk, és bemutatjuk a szükséges matematikai és informatikai eszközöket. Amint már azt az előző fejezetben említettük, a Cournot-féle modellben a vállalatok kibocsátási szinteket határoznak meg és egyazon időben hozzák meg döntéseiket. A modellhez szükséges feltételek:

- a piacon két vállalat van,
- nincs lehetőség további vállalatok belépésére,
- a vállalatok egyforma termékeket termelnek.

3.1. Időben állandó modell

A második fejezetben bevezetett jelöléseket felhasználjuk. Ha a vállalatok által kibocsátott termékek árai azonosak, a modellben az ár (P) és a termelt összmennyiség (Q) a következőképpen alakul:

$$P = P(Q),$$
$$Q = q_1 + q_2,$$

A keresleti súlyozás figyelembe vételével:

$$Q = w\,q_1 + (1 - w)\,q_2 \quad (0 \le w \le 1).$$

A teljes bevételfüggvények ekkor rendre a következő alakúak:

$$\mathrm{TR}_1 = P\,q_1, \qquad \mathrm{TR}_2 = P\,q_2$$

A profitfüggvények:

$$\pi_1 = \mathrm{TR}_1 - \mathrm{TC}_1, \quad \pi_2 = \mathrm{TR}_2 - \mathrm{TC}_2.$$

Azon feltevés szerint, hogy az első cég úgy maximalizálja profitját, hogy közben a második kibocsátása állandó, differenciálhatjuk a π_i profitokat q_i szerint, miközben q_j konstans ($i = 1, 2$; $j = 2, 1$). A teljes- bevétel és költség szintén a kibocsátott mennyiségek függvényeiként írhatók le. A

$$\partial_{q_1} \pi_1 = 0, \quad \partial_{q_2} \pi_2 = 0 \tag{1}$$

egyenleteket megoldva kapjuk az egyes vállalatok reakciógörbéit (maximális profit görbék). A Cournot-féle megoldás, amikor a két reakciógörbe metszi egymást. Ez az a pont, ahol mindkét vállalat profitja maximális.

Vizsgáljuk a modellt egy konkrét példa esetében!

1. PÉLDA

A Cournot-duopólium ezen formájára példa lehet az úgynevezett "*Kólaháború*" a Coca-Cola Company és a PepsiCo Incorporated vállalatok között. A fejezet elején közölt feltételek teljesülnek, mivel

– az üdítőital piacon ez a két vállalat játssza a főszerepet, hiszen együttesen a piac 75%-át lefedik,
– több okból is nehéz más cégek belépése erre a piacra; például mind a Coca-Cola, mind a Pepsi megállapodást kötött palackozóikkal más cégek által előállított hasonló termékek palackozásának korlátozására, mindemellett mindkét cég hosszú múltra tekint vissza, így egy új vállalat belépése erre a piacra két ilyen hatalmas rivális mellett nehéznek mondható,
– a cégek által előállított termékek homogénnek tekinthetők [8].

A két vállalatról feltehetjük, hogy azonos árakon kínálják az egyes termékeket, így a verseny a kibocsátott mennyiségekben nyilvánul meg. Jelölje qc a Coca-Cola Company által, qp pedig a PepsiCo Incorporated által termelt mennyiséget, melyek millió gallon szirupban értendőek. A [8]-ban található példa alapján, a két cég árfüggvényét azonosnak tekintve az alábbi értékekkel dolgozunk:

```
Clear["Global`*"]
Price = 20 - 2 qc - qp;
```

A két vállalat teljes költség- és bevétel függvényei a diétás üdítőital piacon:

```
TCCoke = 5 qc; TCPepsi = 5 qp;
```

```
TRCoke = Price qc; TRPepsi = Price qp;
```

A két vállalat profitja a következőképpen alakul:

```
πCoke = Expand[TRCoke - TCCoke]
```

$$15\,qc - 2\,qc^2 - qc\,qp$$

```
πPepsi = Expand[TRPepsi - TCPepsi]
```

$$15\,qp - 2\,qc\,qp - qp^2$$

A vállalatok reakciógörbéinek egyenletei:

```
solCoke = Simplify[Solve[∂qc πCoke == 0, qc]] [[1, 1]]
```

$$qc \to \frac{15 - qp}{4}$$

```
solPepsi = Simplify[Solve[∂qp πPepsi == 0, qp]] [[1, 1]]
```

$qp \to \frac{15}{2} - qc$

A Cournot-féle megoldás, amikor a két reakciógörbe metszi egymást.

```
Sol =
 Solve[{qc == (qc /. solCoke), qp == (qp /. solPepsi)},
   {qc, qp}] // N
```

```
{{qc → 2.5, qp → 5.}}
```

A Coca Cola-nak 2.5, a Pepsinek 5 millió gallon szirupot kell termelnie. Ekkor a cégek közös profitja 37.5 millió dollár.

```
C1 = πCoke /. Sol[[1]];
C2 = πPepsi /. Sol[[1]];
C1 + C2
```

```
37.5
```

Az inverz aggregált keresleti függvény alapján az egységár:

```
Price /. Sol[[1]]
```

```
10.
```

Ábrázolva a következőt kapjuk (Coca - Cola kék, Pepsi piros görbék):



A reakciófüggvények (piros és kék szaggatott vonalak)
és az egyes vállalatok isoprofit görbéi.

1. ábra

A cégek számára elérhető legnagyobb profitot a monopol helyzet biztosítaná, vagyis amikor a másik cég kibocsátása nulla.

Interaktív kísérletek

A fejezetben bemutatott időben állandó modellhez interaktív illusztráció készült. Az illusztrációban lehetőség van a keresleti súlynak, az árfüggvénynek és a költségfüggvényeknek a beállítására. Ezen paraméterek mellett rajzolja ki a vállalatok árfüggvényét, a költségfüggvényeket, illetve vagy az egyes cégek profitfüggvényeit, vagy a reakciógörbéket, az isoprofit görbéket és a Cournot-megoldást. Ezen felül, ha a Cournot-megoldás értelmezhető, akkor megadja annak pontos helyét, az ott realizálható profitokat a hozzá tartozó eladási árral.

Az interaktív illusztrációkban szereplő példák fiktívek. A programok célja a két vállalatból álló piaci formák vizsgálata, ezért nem tartjuk szükségesnek a valuták és mértékegységek feltüntetését.

A “profitfüggvények” és a “Cournot-megoldás” fül választásával jeleníthetőek meg a vizsgálni kívánt függvények.

▫ *Profitfüggvények fül*

A “Profitfüggvények” fül választása esetén a “Full PlotRange” segítségével állíthatjuk be, hogy az egyes profitokat nemnegatív, vagy automatikusan beállított tartományon ábrázoljuk.

Az ábra kirajzolja az azonosan nulla szintvonalat, továbbá ennek metszetét az aktuális profitfüggvénnyel, melyet az első cég esetében vastag kék, a második cégnél pedig vastag piros görbe mutat.

▫ *Cournot-megoldás fül*

A “Cournot-megoldás” fül esetén pedig ha több egyensúlyi helyzet van, akkor az olvasó maga állíthatja be a számára megfelelő Cournot-megoldást a “Kezdeti mennyiségek” gomb bejelölése után megjelenő csúszkák segítségével. A Cournot-megoldás értékét elegendő közelítően megadni, a program megtalálja a kapott értékekhez tartozó legközelebbi megoldást.

A program a “Nincs Cournot-megoldás!” szöveget írja ki az alábbi esetekben:

– Ha a Cournot-megoldás helyén valamelyik, vagy mindkét cég által kibocsátott mennyiség negatív érték.

- Ha a Cournot-megoldás helyén valamelyik, vagy mindkét cég által kibocsátott mennyiség komplex érték.
- Ha valamelyik vállalat proftitja negatív, vagy komplex az egyensúlyi helyzetben.

A program ezen esetek kizárása után jeleníti meg a reakció függvényeket, az isoprofit görbéket, és a Cournot-megoldást. Az árfüggvényt a költségfüggvényeket, illetve az isoprofit görbéket akkor is kirajzolja, ha a Cournot-megoldás nem létezik. Az első vállalathoz a kék, a másodikhoz piros színű grafikonok tartoznak.


Időben állandó modell: Versengés
Keresleti súly: Q=w q1+(1−w)q2
w1 0.5
Árfüggvény: P(Q)= b − dQ
b 90
d 40
Első vállalat költségfüggvénye:
TC1=FC1+VC1
FC1 13
VC1=a1q1+b1q1c1+1
a1 5
b1 1.5
c1 0.
Második vállalat költségfüggvénye:
TC2=FC2+VC2
FC2 13
VC2=a2q2+b2q2c2+1
a2 5
b2 1.5
c2 0.2
Full PlotRange
Kezdeti mennyiségek
Profitfüggvények
Cournot−megoldás
TC=FC+VC
TC
q
0 2 4 6 8 10
Árfüggvény:P(Q)
P(Q)
80
60
40
20
0
Q
q2
q1
0 1 2 3 4
Verseny esetén
26.1246 + 25.4016 =
51.5261 profit
34.473 eladási áron
1.39865 és 1.3777
mennyiségekkel realizálható.

Speciális esetek

Ahogyan azt már említettük, mindkét vállalat azonos árakon bocsátja piacra a terméket, a verseny a kibocsátott mennyiségek számában folyik. Ezt mennyiségi versenynek nevezzük.

A Cournot-duopólium modellekben a mennyiségi verseny által maximalizálható profit az ár és ezzel együtt a teljes bevétel, illetve a teljes költség függvényében változik. Ha adott a keresleti függvény, akkor ezzel együtt az eladási ár is meghatározott. A vállalatok ekkor a költségeik csökkentésével szeretnék növelni a profitokat. Ilyen költség csökkentő lehetőség a Kutatás - Fejlesztés tevékenysége. Ekkor ugyanis az egyik vállalat technológiai fejlődése hatására az információáramlás és a szellemeti tulajdon védelmének nehézségei következtében a másik vállalat is szert tesz az új információkra, a Kutatás - Fejlesztés eredményekre gyakorlatilag ingyen. Mindezek mellett, ha egy vállalat olyan területen működik, ahol mellette más vállalatok is termelnek az övével homogénnek mondható termékeket, akkor a megfelelő szállítási vagy egyéb szolgáltatásokhoz tartozó struktúra jelenléte csökkentheti a szállítási költségeket.

A költségek csökkentését vizsgáló modellekről bővebbben a [17,18] cikkekben olvashatunk.

1. KÍSÉRLET Lineáris árfüggvény

A termelt mennyiséget a költségfüggvények befolyásolják. Magasabb költségek esetén kevesebbet tud termelni a vállalat, mint az alacsonyabb költség mellett. Így a költségek a kibocsátott mennyiségekre is hatással vannak. Hogyan változnak a profitok, ha az egyik cég képes csökkenteni költségeit?

Vizsgáljuk a modellt, amikor a cégek által termelt mennyiségek egyenlő arányban vannak jelen a piacon! Legyenek a beállítások:

$$
\begin{aligned}
&w = 0.5,\\
&P(Q) = b - d\,Q,\\
&b = 90,\ d = 40,\\
&\mathrm{FC}_1 = 13,\ \mathrm{FC}_2 = 13,\\
&a_1 = 5,\ b_1 = 1.5, c_1 = 0.2,\\
&a_2 = 5,\ b_2 = 1.5,\ c_2 = 0.2.
\end{aligned}
$$

A cégek árfüggvénye és költségfüggvényei azonosak, ezáltal a kibocsátott mennyiségek és a realizált profitok is megegyeznek. Ha most elkezdjük csökkenteni az első vállalat költségeit (pl.: $c_1 = 0$), akkor profitja nagyobb lesz, mint a másik vállalaté. Érdekes módon a $c_1 = 0$, értéket választva az első cég profitja kisebb mértékben nő, mint ahogyan a másodiké csökken, így az iparágban összesen realizálható profit kevesebb lesz, mint azonos költségfüggvények esetén.

2. KÍSÉRLET Hiperbolikus árfüggvény

Vegyük az alábbi alapbeállításokat:

$w = 0.5,$
$P(Q) = d\,Q^{-\epsilon},$
$d = 40,\ \epsilon = 1,$
$\mathrm{FC}_1 = 13,\ \mathrm{FC}_2 = 13,$
$a_1 = 5,\ b_1 = 1.5,\ c_1 = 0.2,$
$a_2 = 5,\ b_2 = 1.5,\ c_2 = 0.2.$

Az első vállalat a Kutatás és Fejlesztés területbe fektetve új technológiát valósít meg, melynek következtében állandó költségei nőnek $\mathrm{FC}_1 = 15$, változó költségei viszont lényegesen lecsökkennek $c_1 = 0$. A fejlesztésekkel kapcsolatos információt valahogyan a másik vállalat is megszerzni, így az állandó költségei változtatása nélkül $\mathrm{FC}_2 = 13$ képes lecsökkenteni változó költségeit $c_2 = 0.1$. Ekkor az egyensúlyban realizálható profitok mindkét vállalat esetében lecsökkennek, viszont a második cég több profitot realizál, mint az első. Az egyensúlyi ár szintén lecsökken, ami a fogyasztók számára kedvezőnek mondható. Ha a termék iránti kereslet árrugalmatlan, például $\epsilon = 0.5$, akkor lényegesen magasabb profit realizálható mindkét cég számára a kísérlet beállításai mellett. Ezekután az új technológia bevezetése növeli az első és csökkenti a második vállalat profitját. Az eladási ár kisebb, mint az egységnyi árrugalmasságú kereslet esetén, és ez tovább csökken az új technológiának köszönhetően. Az árrugalmas kereslet esetén $(\mathrm{pl.}: \epsilon = 1.125)$ az egységnyi rugalmasságúnál kicsivel magasabb, de a rugalmatlannál alacsonyabb profitszintek valósíthatók meg az egyensúlyban. A vevők szemponjából az árrugalmatlan kereslet esetén a legkedvezőbb az egyensúlyi ár.

Lineáris árfüggvény esetén az új technológia bevezetése mindkét vállalat szempontjából kedvezőtlennek mondható. Ez a fajta árképzés sokkal magasabb egyensúlyi árat eredményez, mint a hiperbolikus árfüggvény esetén tapasztalható.

3.2. Időben diszkrét modell

Ebben a részben az előbb ismertetett időben állandó modellt kiegészítjük azzal a feltevéssel, hogy minden időszaki periódus során a vállalatok visszaemlékeznek saját, illetve a másik válallat gazdasági döntéseire. Mindkét vállalat a t-edik időpontban a $[t,\ t+1]$ szakaszra nézve próbálja profitját maximalizálni. Számukra a rivális vállalat $[t-1,\ t]$ intervallumbeli kibocsátási szintje az utolsó információ, ezért fel kell tételezniük, hogy a rivális a $[t,\ t+1]$ periódusban a $[t-1,\ t]$

periódussal megegyező kibocsátási szintet választ [2].

Ennek megfelelően a profitfüggvények $\pi_1(q_1(t), q_2(t-1))$ és $\pi_2(q_2(t-1), q_2(t))$ alakúak lesznek. A módszer az előző pontban ismertettek alapján történik, a profitok:

$$\pi_1(q_1(t), q_2(t-1)) =$$
$$\mathrm{TR}_1(q_1(t), q_2(t-1)) - \mathrm{TC}_1(q_1) = P(q_1(t), q_2(t-1))\, q_1(t) - \mathrm{TC}_1(q_1(t)),$$
$$\pi_2(q_1(t-1), q_2(t)) =$$
$$\mathrm{TR}_2(q_1(t-1), q_2(t)) - \mathrm{TC}_2(q_2) = P(q_1(t-1), q_2(t))\, q_2(t) - \mathrm{TC}_2(q_2(t)).$$

A profitok differenciálásával kapott

$$\frac{\partial}{\partial q_1(t)}\pi_1(q_1(t), q_2(t-1)) = 0, \quad \frac{\partial}{\partial q_2(t)}\pi_2(q_1(t-1), q_2(t)) = 0 \tag{2}$$

egyenletek adják az egyes vállalatok reakciógörbéit. A (2) az alábbi ekvivalens alakban is felírható:

$$P(q_1(t), q_2(t-1)) = \frac{\partial}{\partial q_1(t)}\mathrm{TC}_1(q_1(t)), \quad P(q_1(t-1), q_2(t)) = \frac{\partial}{\partial q_2(t)}\mathrm{TC}_2(q_2(t)).$$

Az első egyenletből $q_1(t)$-t a másodikból $q_2(t)$-t kifejezve, a

$$q_1(t) = f(q_2(t-1)), \quad q_2(t) = g(q_1(t-1)) \tag{3}$$

differenciaegyenlet-rendszert kapjuk a $q_1(0)$, $q_2(0)$, kezdeti értékekkel, ahol f és g a másik cég által előző periódusban kibocsátott mennyiségek valamely függvényei.

Az időben állandó modell Cournot-megoldása a (3) dinamikai rendszer fixpontja. Ismert, hogy ennek stabilitása a rendszer Jacobi mátrixának segítségével vizsgálható. Ha a Jacobi mátrix minden sajátértékének abszolút értéke kisebb mint egy, akkor a megoldás aszimptotikusan stabilis [12]. A stabilitás vizsgálatát egy példán keresztül szemléltetjük, a bonyolultabb esetekben a modellhez készült interaktív illusztráció eredményeire hagyatkozunk.

2. PÉLDA

Módosítjuk a fejezet 1. példáját a modellnek megfelelően:

```
Clear[Price, TCCoke, TCPepsi, TRCoke, TRPepsi,
 πCoke, πPepsi, solCoke, solCokem, solPepsi]
```

A cégek árfüggvénye:

```
Price = 20 - 2 qc - qp;
```

A költségfüggvények és a bevétel függvények:

```
TCCoke = 5 qc; TCPepsi = 5 qp;
```

```
TRCoke = Price qc; TRPepsi = Price qp;
```

Feltételezzük, hogy a rivális vállalat az előző periódussal megegyező mennyiséget bocsájt ki, ezáltal a profitfüggvények:

```
πCoke = Expand[TRCoke - TCCoke]
```

$15\,qc - 2\,qc^2 - qc\,qp$

```
πPepsi = Expand[TRPepsi - TCPepsi]
```

$15\,qp - 2\,qc\,qp - qp^2$

A két cég profitfüggvényeinek maximuma ekkor a következőképpen alakul:

```
solCoke = Simplify[Solve[D[πCoke, qc] == 0, qc]][[1, 1]]
```

$qc \to \frac{15 - qp}{4}$

```
solPepsi = Simplify[Solve[D[πPepsi, qp] == 0, qp]][[1, 1]]
```

$qp \to \frac{15}{2} - qc$

Eszerint a $(t+1)$-dik intervallumban

$$q_c(t+1) = \frac{15 - q_p(t)}{4},$$
$$q_p(t+1) = \frac{15}{2} - q_c(t).$$

Ennek a dinamikus rendszernek a fixpontja adja a Cournot-megoldást, ami azonos az időben állandó modell megoldásával.

```
Solve[
 Map[# /. Rule → Equal &, {solCoke, solPepsi}], {qc, qp}]
```

$\left\{\left\{qc \to \frac{5}{2},\ qp \to 5\right\}\right\}$

Vizsgáljuk meg a kibocsátott mennyiségek változását konkrét kezdeti értékek esetén! Nézzük meg hogyan viselkedik a rendszer tetszőleges különböző kezdeti értékekből indulva.

```
Clear[DESolve];
DESolve[f_, IC_?MatrixQ, n_Integer] :=
 Map[NestList[f, #, n] &, IC]
```

```
var = {qc, qp};
f[{t_, qc_, qp_}] =
 {t + 1, qc, qp} /. {solCoke, solPepsi}
```

$$\left\{1+t,\ \frac{15-\text{qp}}{4},\ \frac{15}{2}-\text{qc}\right\}$$

A $t = 0$ időpontból indulva több kezdeti értékekből $\{q_c(0),\ q_p(0)\}$indítunk megoldásokat:

```
IC = {{0., 0., 0.}, {0., 2., 0.5}, {0., 4., 8.},
   {0., 8., 1.}, {0., 2., 2.}, {0., 3., 7.}};
```

```
NN = 6; sol = DESolve[f, IC, NN];
```

A megoldásokat ábrázolva láthatjuk, hogy a kezdeti értékektől függetlenül a megoldások az egyensúlyi helyzethez tartanak.



2. ábra

Megjegyezzük, hogy ebben az esetben a qc[t], qp[t] függvények az RSolve utasítás segítségével explicit módon is megadhatók:

```
FullSimplify[
 RSolve[{qc[t] == (solCoke[[2]] /. {qp → qp[t - 1]}),
   qp[t] == (solPepsi[[2]] /. {qc → qc[t - 1]}),
```

```
   qc[0] == qc0, qp[0] == qp0}, {qc[t], qp[t]}, t]]
```

$$\left\{\left\{qc[t] \to \frac{5}{2} + 2^{-2-t}\left(2\,qc0 - qp0 + (-1)^{t}\,(-10 + 2\,qc0 + qp0)\right),\; qp[t] \to 5 + 2^{-1-t}\left(-2\,qc0 + qp0 + (-1)^{t}\,(-10 + 2\,qc0 + qp0)\right)\right\}\right\}$$

Ellenőrizzük az egyensúlyi helyzet stabilitását (3) jobb oldala Jacobi mátrixának sajátértékei vizsgálatával. Ha ezek a sajátértékek a Cournot-megoldás helyén véve egynél kisebb abszolút értékűek, akkor a fix pont aszimptotikusan stabilis [19].

Valóban, felírva a (3) Jacobi mátrixát:

```
JacobiM =
  D[FullSimplify[{solCoke[[2]] /. {qp[t - 1] → qp},
     solPepsi[[2]] /. {qc[t - 1] → qc}}], {{qc, qp}}];
MatrixForm[JacobiM]
```

$$\begin{pmatrix} 0 & -\frac{1}{4} \\ -1 & 0 \end{pmatrix}$$

A sajátértékek :

```
Eigenvalues[JacobiM]
```

$$\left\{-\frac{1}{2}, \frac{1}{2}\right\}$$

abszolút értéke kisebb mint egy, így a rendszer fixpontja $q_c = 2.5$, $q_p = 5$ aszimptotikusan stabilis.

Interaktív kísérletek

Az időben diszkrét rendszerhez is készült interaktív illusztráció. Lehetőség van a keresleti súly, az árfüggvény és annak paraméterei, továbbá a költségfüggvények beállítására. Az iterációs lépés számát és a kezdeti értékek maximumait is a felhasználó választhatja meg. Ezek alapján a program kirajzolja az egyes költségfügg-vényeket, az árfüggvényt, továbbá a diszkrét idejű megoldást, illetve kiírja a Cournot-megoldás helyét, ha az létezik.

Időben diszkrét modell: Versengés

Keresleti súly: $Q = w\, q_1 + (1-w) q_2$

w 0.5

Árfüggvény: P(Q)= $b - dQ$

b 90 d 40

Első vállalat költségfüggvénye:

$TC_1 = FC_1 + VC_1$

FC_1 13

$VC_1 = a_1 q_1 + b_1 q_1^{c_1+1}$

a_1 5

b_1 1.5

c_1 0.

Második vállalat költségfüggvénye:

$TC_2 = FC_2 + VC_2$

FC_2 13

$VC_2 = a_2 q_2 + b_2 q_2^{c_2+1}$

a_2 5

b_2 1.5

c_2 0.2

Iteráció 5

Kezdeti értékek:

$q_{1,min}$ 1 $q_{2,min}$ 1

$q_{1,max}$ 5 $q_{2,max}$ 5







Egyensúlyi helyzet:

(1.39865, 1.3777)

$\pi_1 = 26.1246$

$\pi_2 = 25.4016$

Speciális esetek

3. KÍSÉRLET Lineáris árfüggvény

A 1. kísérletnek megfelelően vizsgáljuk a diszkrét idejű modellt! Ekkor a cégek által termelt mennyiségek egyenlő arányban vannak jelen a piacon. A beállítások:

$$P(Q) = b - d\,Q,$$
$$b = 90,\ \ d = 40,$$
$$\mathrm{FC}_1 = 13,\ \mathrm{FC}_2 = 13,$$
$$a_1 = 5,\ b_1 = 1.5,\ c_1 = 0.2$$
$$a_2 = 5,\ b_2 = 1.5,\ c_2 = 0.2.$$

Az interaktív ábra jól mutatja, hogy ekkor a különböző kezdeti értékekből induló megoldások a Cournot egyensúlyi helyzet felé tartanak, tehát megállapíthatjuk, hogy az ábra szerint az egyensúlyi helyzet ekkor aszimptotikusan stabil. Változtatva a keresleti súlyt, a stabilitást illetően nem tapasztalunk változást. Az árfüggvény paramétereinek változtatása a Cournot-megoldás létezését befolyásolja. A költségek változtatása szintén nincs hatással a megoldás stabilitására, annak csupán helyét és értékét befolyásolja. Megállapíthatjuk, hogy ezen árképzés esetén a megoldás, ha létezik, akkor aszimptotikusan stabil lesz.

4. KÍSÉRLET Hiperbolikus árfüggvény

Javasoljuk az alábbi beállítások esetén a megoldás stabilitásának vizsgálatát:

$$w = 0.5,$$
$$P(Q) = d\,Q^{-\epsilon},$$
$$d = 40,\ \epsilon = 1,$$
$$\mathrm{FC}_1 = 13,\ \mathrm{FC}_2 = 13,$$
$$a_1 = 5,\ b_1 = 1.5,\ c_1 = 0.2,$$
$$a_2 = 5,\ b_2 = 1.5,\ c_2 = 0.2.$$

A megoldás ekkor szintén aszimptotikusan stabilis. Ebben az esetben a keresleti súlyt változtatva (pl.: $w = 0.7$), a megoldás elveszíti stabilitását. A költségeket változtatva nem tapasztalunk változást, a kezdeti beállítások mellett a megoldás továbbra is aszimptotikusan stabil marad. Kis ϵ esetén (pl.: $\epsilon = 0.05$) a megoldás nem létezik, nagy ϵ esetén ($\epsilon > 1$) a megoldás instabillá válhat.

3.3. Időben folytonos modell

Ebben a részben a modellt folytonos időben vizsgáljuk. A vállalatok folyamatosan figyelik a piac alakulását és a rivális tevékenységét. A bevezetett jelöléseket alkalmazva a teljes bevételek:

$$\mathrm{TR}_1 = P\,q_1, \quad \mathrm{TR}_2 = P\,q_2.$$

A teljes költségek:

$$\begin{aligned}\mathrm{TC}_1 &= \mathrm{AC}_1(q_1)\,q_1 = \mathrm{FC}_1 + \mathrm{VC}_1(q_1) = \mathrm{FC}_1 + a_1\,q_1 + b_1\,q_1^{c_1+1},\\ \mathrm{TC}_2 &= \mathrm{AC}_2(q_2)\,q_2 = \mathrm{FC}_2 + \mathrm{VC}_2(q_2) = \mathrm{FC}_2 + a_2\,q_2 + b_2\,q_2^{c_2+1}.\end{aligned}$$

Ezek alapján a profitok:

$$\pi_1 = \mathrm{TR}_1 - \mathrm{TC}_1, \quad \pi_2 = \mathrm{TR}_2 - \mathrm{TC}_2.$$

Ezek után azt feltételezzük, hogy a kibocsátás tetszőleges, folytonosan konstans hányada a kívánt és az aktuális kibocsátási szint különbségének, vagyis

$$\begin{aligned} q_1'(t) &= k_1(x_1(t) - q_1(t)),\\ q_2'(t) &= k_2(x_2(t) - q_2(t)), \end{aligned} \tag{4}$$

ahol x_1 és x_2 a kívánt kibocsátási szintek, $k_1, k_2 > 0$ konstans értékek. Ekkor viszont tudjuk, hogy ezeken a szinteken az egyes vállalatok profitja maximális, feltéve hogy a másik vállalat nem változtat a kibocsátáson. Tehát $x_1(t)$ és $x_2(t)$ az

$$\begin{aligned} \frac{\partial}{\partial q_1(t)}\pi_1(t) &= 0,\\ \frac{\partial}{\partial q_2(t)}\pi_2(t) &= 0, \end{aligned} \tag{5}$$

egyenletek megoldásai.

Az egyszerűség kedvéért a modell stabilitását csak bizonyos esetben vizsgáljuk.

3.3.1 Lineáris ár, lineáris, vagy négyzetes költségek

Az árfüggvény a következő alakú:

$$P(Q) = b - d(w\,q_1 + 1 - w\,q_2).$$

Ha a költségfüggvényekben a c_1 és c_2 értékeket nullának, vagy egynekválasszuk, akkor a költségfüggvények ettől függően lineárisak, vagy négyzetesek lesznek az adott vállalat által termelt mennyiség függvényében.

Ekkor a (4) differenciálegyenlet-rendszer a következő általános alakban írható fel:

$$\begin{aligned} q_1'(t) &= k_1(a\,q_2(t) + b - q_1(t))\\ q_2'(t) &= k_2(c\,q_1(t) + d - q_2(t)), \end{aligned} \tag{6}$$

vagy mátrix alakban kifejezve:

$$\begin{pmatrix} q_1'(t) \\ q_2'(t) \end{pmatrix} = \begin{pmatrix} -k_1 & k_1\,a \\ k_2\,c & -k_2 \end{pmatrix} . \begin{pmatrix} q_1(t) \\ q_2(t) \end{pmatrix} + \begin{pmatrix} k_1\,b \\ k_2\,d \end{pmatrix}.$$

A várt kibocsátások (5) miatt a másik vállalat kibocsátásának függvényeként írhatók le $(x_1(t) = a\,q_2(t) + b,\ \ x_2(t) = c\,q_1(t) + d)$, ahol a, b, c és d konstans értékek. A rendszer egyensúlyi helyzetei a

$$k_1(a\,q_2 + b - q_1) = 0,$$
$$k_2(c\,q_1 + d - q_2) = 0$$

egyenletek megoldásai:

$$q_1 = -\frac{b + a\,d}{-1 + a\,c},\qquad q_2 = -\frac{b\,c + d}{-1 + a\,c} \tag{7}$$

A (6) differenciálegyenlet-rendszer állandó együtthatós, lineáris és homogén, ha $b = d = 0$, inhomogén ha $b \neq 0$ és $d \neq 0$. Az egyensúlyi helyzetek stabilitása (6) rendszer együtthatómátrixának sajátértékei határozzák meg.

1. TÉTEL

Tegyük fel, hogy $k_1 > 0$, $k_2 > 0$ továbbá a,b, c és d valós számok. Ekkor a (7) által meghatározott egyensúlyi helyzet stabilitását az alábbi módon vizsgálhatjuk.

A $k_1 = k_2$ eset.

– *Ha $a\,c < 1$, akkor az egyensúlyi helyzet aszimptotikusan stabilis.*
– *Ha $a\,c = 1$, akkor nem tudunk semmit az egyensúlyi helyzet stabilitásáról.*
– *Ha $a\,c > 1$, akkor az egyensúlyi helyzet instabil.*

A $k_1 \neq k_2$ eset.

– *Ha $4\,a\,c + \frac{k_1}{k_2} + \frac{k_2}{k_1} < 2$, akkor az egyensúlyi helyzet aszimptotikusan stabilis.*
– *Ha $4\,a\,c + \frac{k_1}{k_2} + \frac{k_2}{k_1} \geq 2$, akkor az alábbi aleseteket különböztetjük meg:*

- *$a\,c < 1$ esetén az egyensúlyi helyzet aszimptotikusan stabilis,*
- *$a\,c = 1$ esetén a stabilitásról nem tudunk semmit,*
- *$a\,c > 1$ esetén pedig az egyensúlyi helyzet instabil.*

BIZONYÍTÁS

A (6) egyenlet jobb oldalához tartozó mátrix:

```
EhMtx = {{-k1, a k1}, {c k2, -k2}};
```

```
MatrixForm[EhMtx]
```

$$\begin{pmatrix} -k_1 & a\,k_1 \\ c\,k_2 & -k_2 \end{pmatrix}$$

Ennek a sajátértékei:

```
λ1 = Eigenvalues[EhMtx][[1]]
```

$$\frac{1}{2}\left(-k_1 - k_2 - \sqrt{\left(k_1^2 - 2\,k_1\,k_2 + 4\,a\,c\,k_1\,k_2 + k_2^2\right)}\right)$$

```
λ2 = Eigenvalues[EhMtx][[2]]
```

$$\frac{1}{2}\left(-k_1 - k_2 + \sqrt{\left(k_1^2 - 2\,k_1\,k_2 + 4\,a\,c\,k_1\,k_2 + k_2^2\right)}\right)$$

Ha $k_1 = k_2 = k > 0$, akkor a sajátértékek

```
Simplify[λ1 /. {k1 → k, k2 → k}]
```

$$-k - \sqrt{a\,c\,k^2}$$

```
Simplify[λ2 /. {k1 → k, k2 → k}]
```

$$-k + \sqrt{a\,c\,k^2}$$

alakúak. Az egyik sajátérték mindig negatív valós részű.

A másik sajátérték, ha $a\,c < 0$, akkor negatív valósrészű komplex szám. Ha $a\,c = 0$, akkor $\lambda_1 = \lambda_2 < 0$ valós érték. Ha $0 < a\,c < 1$, akkor $\lambda_2 \neq \lambda_1$ sajtáérték valós. Ha $a\,c = 1$, akkor $\lambda_2 = 0$. Ha pedig $a\,c > 1$, akkor $\lambda_2 > 0$ pozitív valós szám.

Ha $k_1 \neq k_2$, k_1, $k_2 > 0$, akkor az egyik sajátérték, λ_1 szintén mindig negatív valósrészű. Az egyensúlyi helyzet stabilitásához λ_2 sajátértéknek is negatív valósrészűnek kell lennie.

– $k_1^2 + 2\,(-1 + 2\,a\,c)\,k_1\,k_2 + k_2^2 < 0$, ekkor mindkét gyök negatív valósrészű komplex szám.

```
Simplify[Reduce[
  k1^2 + 2 (-1 + 2 ac) k1 k2 + k2^2 < 0 && k1 > 0 && k2 > 0, {ac}]]
```

$$k_1 > 0\ \&\&\ k_2 > 0\ \&\&\ 4\,ac + \frac{k_1}{k_2} + \frac{k_2}{k_1} < 2$$

A $k_1^2 + 2\,(-1 + 2\,a\,c)\,k_1\,k_2 + k_2^2 \geq 0$, esetben az alábbi aleseteket különböztethetjük meg.

– $\sqrt{k_1^2+2(-1+2\,a\,c)\,k_1\,k_2+k_2^2} < k_1+k_2$. Ebben az esetben a sajátértékek negatív valós számok. Ha a gyök alatt álló kifejezés nullával egyenlő, akkor a két sajátérték egybeesik.

```
Simplify[Reduce[√(k1^2 + 2 (-1 + 2 ac) k1 k2 + k2^2) < k1 + k2 &&
    k1^2 + 2 (-1 + 2 ac) k1 k2 + k2^2 ≥ 0 && k1 > 0 && k2 > 0, {ac}]]
```

$k_1 > 0$ && $k_2 > 0$ && $4\,ac + \frac{k_1}{k_2} + \frac{k_2}{k_1} \geq 2$ && $ac < 1$

– $\sqrt{k_1^2+2(-1+2\,a\,c)\,k_1\,k_2+k_2^2} = k_1+k_2$ esetén λ_2 nulla valósrészű.

```
Simplify[Reduce[√(k1^2 + 2 (-1 + 2 ac) k1 k2 + k2^2) == k1 + k2 &&
    k1^2 + 2 (-1 + 2 ac) k1 k2 + k2^2 ≥ 0 && k1 > 0 && k2 > 0, {ac}]]
```

$k_1 > 0$ && $k_2 > 0$ && $ac == 1$

– $\sqrt{k_1^2+2(-1+2\,a\,c)\,k_1\,k_2+k_2^2} > k_1+k_2$. Ekkor $\lambda_2 > 0$ pozitív valós szám.

```
Simplify[Reduce[√(k1^2 + 2 (-1 + 2 ac) k1 k2 + k2^2) > k1 + k2 &&
    k1^2 + 2 (-1 + 2 ac) k1 k2 + k2^2 ≥ 0 && k1 > 0 && k2 > 0, {ac}]]
```

$k_1 > 0$ && $k_2 > 0$ && $ac > 1$

Ezzel tehát a tételt beláttuk.

■

3. PÉLDA

Az 1. példát most folytonos időben vizsgáljuk. A diszkrét esettel ellentétben a cégek kibocsátásait ekkor nem időszakokra, hanem minden egyes időpillanatra meghatározzuk.

```
Clear[Price, TCCoke, TCPepsi,
  TRCoke, TRPepsi, πCoke, πPepsi]
```

Ennek megfelelően az árfüggvény:

```
Price[t] = 20 - 2 qc[t] - qp[t];
```

A költség- és bevételfüggvények:

```
TCCoke[t] = 5 qc[t];
TCPepsi[t] = 5 qp[t];
```

```
TRCoke[t] = Price[t] qc[t];
```

```
TRPepsi[t] = Price[t] qp[t];
```

Az egyes cégek profitjai:

```
πCoke[t] = Expand[TRCoke[t] - TCCoke[t]]
```

$$15\,qc[t] - 2\,qc[t]^2 - qc[t]\,qp[t]$$

```
πPepsi[t] = Expand[TRPepsi[t] - TCPepsi[t]]
```

$$15\,qp[t] - 2\,qc[t]\,qp[t] - qp[t]^2$$

A differenciálegyenlet-rendszer:

```
qc'[t] = kc (xc[t] - qc[t]);
qp'[t] = kp (xp[t] - qp[t]);
```

Mivel a kívánt kibocsátási szint mindkét cég esetében maximalizálja a profitot és a másik cég nem változtatja kibocsátását a várt kibocsátások:

```
xc[t] =
 qc[t] /. Simplify[Solve[∂qc[t] πCoke[t] == 0, qc[t]]][[1]]
```

$$\frac{1}{4}\,(15 - qp[t])$$

```
xp[t] = qp[t] /.
  Simplify[Solve[∂qp[t] πPepsi[t] == 0, qp[t]]][[1]]
```

$$\frac{15}{2} - qc[t]$$

Ezt behelyettesítve az előbbi egyenletrendszerbe, az a és a c együtthatókra kapjuk:

```
Coefficient[qc'[t] / kc, qp[t]]
Coefficient[qp'[t] / kp, qc[t]]
```

$$-\frac{1}{4}$$

$$-1$$

Ekkor $a\,c = \frac{1}{4} < 1$. Válasszuk a k értékeket a következő módon: $k_1 = k_2 = k$, amiről tudjuk, hogy pozitív. Az 1. Tétel értelmében az egyensúlyi helyzet aszimptotikusan stabilis. Vizsgáljuk ezt grafikusan!

```
Clear[Eqn2D, Rhs2D, ICeqn, Eh, K, sol]
```

A differenciálegyenlet-rendszer kezdeti értékei legyenek: $q_c(0) = 3$ és $q_p(0) = 7$.

```
Eqn2D = {
   qC'[t] == k (-qC[t] + 1/4 (15 - qP[t])),
   qP'[t] == k (15/2 - qC[t] - qP[t])};
Rhs2D = {k (-qC + 1/4 (15 - qP)), k (15/2 - qC - qP)};
ICeqn = {qC[0] == 3, qP[0] == 7};
```

Az egyensúlyi helyzetek:

```
Eh = Solve[Rhs2D == {0, 0}, {qC, qP}]
```

$\left\{\left\{qC \to \frac{5}{2},\ qP \to 5\right\}\right\}$

A k konstans változtatása a rendszer viselkedését nem befolyásolja. A differenciálegyenlet megoldása $k = 0.5$ és $k = 248$ esetén:

```
K = {0.5, 200};
```

```
sol[t] = {qC[t], qP[t]} /.
   Table[NDSolve[Join[Eqn2D, ICeqn] /. {k → K[[i]]},
     {qC[t], qP[t]}, {t, 0, 5}], {i, 1, 2}];
```



Az iránymező és a megoldás $k = 0.5$ és $k = 200$ esetén.

3. ábra

Az 1. Tételben szereplő a és c konstansok nemnegatív valós számok, (6) együtt-hatómátrixának sajátértékei negatív valós számok, így a stabil egyensúlyi helyzet “csomó”.

Interaktív kísérletek

A modellhez ebben az esetben is készült interaktív illusztráció. Az illusztrációban a keresleti súly, az árfüggvény és a vállalatok költségfüggvényeinek paraméterei, továbbá a reagálás gyorsaságát befolyásoló konstansok változtathatóak. Az ábra kirajzolja a vállalatok költségfüggvényeit, az alkalmazott árfüggvényt, továbbá a vektormezőt és a Cournot-megoldáshoz tartozó reakciógörbéket.

Időben folytonos modell: Versengés

Keresleti súly: $Q = w\, q_1 + (1-w) q_2$

w 0.5

Árfüggvény: $P(Q) =$ $d\, Q^{-\epsilon}$

d 40 ϵ 1

Első vállalat költségfüggvénye: Második vállalat költségfüggvénye:

$TC_1 = FC_1 + VC_1$ $TC_2 = FC_2 + VC_2$

FC_1 13 FC_2 13

$VC_1 = a_1 q_1 + b_1 q_1^{c_1+1}$ $VC_2 = a_2 q_2 + b_2 q_2^{c_2+1}$

a_1 4.6 a_2 5

b_1 1.5 b_2 1.5

c_1 0. c_2 0.115

Reagálás gyorsasága:

k_1 0.5 k_2 0.5

Megoldástól való távolság:

q_1 5 q_2 5

TC=FC+VC



Árfüggvény:P(Q)





Egyensúlyi helyzet:

(3.26657, 2.89213)

$\pi_1 = 9.50587$

$\pi_2 = 5.20569$

Speciális esetek

5. KÍSÉRLET Lineáris árfüggvény

Az 1. kísérlet paramétereivel vizsgáljuk a Cournot-megoldás stabilitását!

$w = 0.5$,
$P(Q) = b - d\,Q$,
$b = 90$, $d = 40$,
$\mathrm{FC}_1 = 13$, $\mathrm{FC}_2 = 13$,
$a_1 = 5$, $b_1 = 1.5$, $c_1 = 0.2$,
$a_2 = 5$, $b_2 = 1.5$, $c_2 = 0.2$,
$k_1 = 0.5$, $k_2 = 0.5$.

Az ábra alapján azt mondhatjuk, hogy az egyensúlyi helyzet stabilis. Az 1. kísérletet elvégezve az időben folytonos modellen, láthatjuk, hogy a megoldás továbbra is stabil marad. Sem a keresleti súly, sem a költségek változtatása nem befolyásolja a megoldás stabilitását.

6. KÍSÉRLET Hiperbolikus árfüggvény

Tekintsük a következő beállításokat:

$w = 0.5$,
$P(Q) = d\,Q^{-\epsilon}$,
$d = 40$, $\epsilon = 1.$,
$\mathrm{FC}_1 = 13$, $\mathrm{FC}_2 = 13$,
$a_1 = 5$, $b_1 = 1.5$, $c_1 = 0.2$,
$a_2 = 5$, $b_2 = 1.5$, $c_2 = 0.2$.

A megoldás aszimptotikusan stabil. Sem a költségfüggvények sem a keresleti súly változtatása nem befolyásolja a megoldás stabilitását. Vizsgáljuk a modellt az ϵ paraméter változtatása mellett! A c_1-et például nullának választva egy elegendően nagy ϵ mellett (pl.: $\epsilon = 1.3$) a megoldás instabil. Vagyis az árrugalmas kereslet alacsony költségek esetén instabil megoldást produkál.